

SAS® Dates with Decimal Time

John Henry King, Ouachita Clinical Data Services, Inc., Caddo Gap, AR

ABSTRACT

Did you know that a SAS-Date can have a decimal part that SAS treats as you might expect a fraction of a day also known as time? Often when data are imported from EXCEL an excel date-time will be imported as SAS-Date with a decimal part. This Rapid Fire tip discusses how to identify these values and gives examples of how to work with them including how to convert them to SAS-Date-time.

INTRODUCTION

A SAS-Date is defined as an integer number representing the number of days since 01JAN1960. For 01JAN1960 the internal SAS Date value is zero. The numeric value stored as Real Binary Floating Point can include a decimal portion and this decimal portion can represent time. For example the SAS-Date 0.5 can be considered to represent ½ day since 01JAN1960 or 01JAN1960:12:00:00. It is important to remember that for the most part SAS ignores the decimal portion and a SAS Date with a decimal value is NOT a SAS-Date-time value.

USUALLY IGNORED

As shown by this example SAS usually ignores the decimal part. Notice that for SAMEDAY alignment INTNX keeps the decimal part while alignment END and BEGIN drop it.

```
data _null_;
  x = .5;
  put x= x=date9.;
  do align='B','E','S';
    i = intnx('week',x,0,align);
    put i= i=date9. i=weekdate. align=;
  end;
run;
```

```
x=0.5 x=01JAN1960
i=-5 i=27DEC1959 i=Sunday, December 27, 1959 align=B
i=1 i=02JAN1960 i=Saturday, January 2, 1960 align=E
i=0.5 i=01JAN1960 i=Friday, January 1, 1960 align=S
```

CONVERSION TO DATE-TIME

Converting a SAS-Date with decimal time part to SAS-Date-Time can be done using DHMS function specifying 0 for the hour, minute, and second arguments.

```
data _null_;
  x = .51;
  put x= x=date9.;
  dt = dhms(x,0,0,0);
  put dt= dt=datetime20.;
run;
```

```
x=0.51 x=01JAN1960
```

```
dt=44064 dt=01JAN1960:12:14:24
```

Interestingly I cannot find mention of this in the SAS documentation for [DHMS function](#).

IMPORT FROM EXCEL XLS

Consider the Excel spread sheet shown in Figure 1 where columns B and C are Excel date-time variables. The underlying values are show in column D and E and are displayed simply to show the “internal” values of the data stored in columns B and C. Excel uses a date system similar to SAS with a different base date.

	A	B	C	D	E
1	Day	Start Time	End Time	Start Time2	End Time2
2	Mon-Wed	5/18/15 8:00	5/20/15 10:30	42142.33333	42144.4375
3	Mon-Wed	5/18/15 8:00	5/20/15 10:30	42142.33333	42144.4375
4	Mon-Wed	5/18/15 8:00	5/20/15 10:30	42142.33333	42144.4375
5	Tuesday	5/19/15 14:45	5/19/15 15:15	42143.61458	42143.63542
6	Monday	5/18/15 10:15	5/18/15 11:05	42142.42708	42142.46181
7	Monday	5/18/15 13:15	5/18/15 14:45	42142.55208	42142.61458
8	Monday	5/18/15 13:15	5/18/15 13:25	42142.55208	42142.55903
9	Mon-Wed	5/18/15 8:00	5/20/15 10:30	42142.33333	42144.4375

Figure 1, Excel file with "excel" date time fields B and C

When this spread sheet is imported into SAS using PROC IMPORT column B and C are converted to SAS dates and the decimal value is retained Figure 2. You won't see the decimal part as the format associated by PROC FORMAT is MMDDYY.

Variables in Creation Order						
#	Variable	Type	Len	Format	Informat	Label
1	Day	Char	12	\$12.	\$12.	Day
2	Start Time	Num	8	MMDDYY10.		Start Time
3	End Time	Num	8	MMDDYY10.		End Time
4	Start Time2	Num	8	BEST12.		Start Time2
5	End Time2	Num	8	BEST18.		End Time2

Page Break

Obs	Day	Start Time	End Time	Start Time2	End Time2
1	Mon-Wed	05/18/2015	05/20/2015	42142.333333	42144.4375
2	Mon-Wed	05/18/2015	05/20/2015	42142.333333	42144.4375
3	Mon-Wed	05/18/2015	05/20/2015	42142.333333	42144.4375
4	Tuesday	05/19/2015	05/19/2015	42143.614583	42143.6354166666
5	Monday	05/18/2015	05/18/2015	42142.427083	42142.4618055555
6	Monday	05/18/2015	05/18/2015	42142.552083	42142.6145833333
7	Monday	05/18/2015	05/18/2015	42142.552083	42142.5590277777
8	Mon-Wed	05/18/2015	05/20/2015	42142.333333	42144.4375

Figure 2, SAS data imported from Excel ‘Start Time’n and “End Time”n converted to SAS date by PROC IMPORT.

If we print the data and remove the formats the internal values are show Figure 3. The data were imported and formatted as DATE and the decimal time portion that EXCEL uses for data-time is retained. This is good as we don't misplace any data but it can cause a problem if we are not aware. For example if we write an expression

```
where 'start time'n eq '18may2015'd;
```

to subset our data we don't get any observations, because 'Start Time'n has a decimal part and that decimal part is not present in the date literal '18may2015'd. We can use the INT function to fix it.

```
where int('start time'n) eq '18may2015'd;
```

Obs	Day	Start Time	End Time	Start Time2	End Time2
1	Mon-Wed	20226.33	20228.44	42142.33	42144.44
2	Mon-Wed	20226.33	20228.44	42142.33	42144.44
3	Mon-Wed	20226.33	20228.44	42142.33	42144.44
4	Tuesday	20227.61	20227.64	42143.61	42143.64
5	Monday	20226.43	20226.46	42142.43	42142.46
6	Monday	20226.55	20226.61	42142.55	42142.61
7	Monday	20226.55	20226.56	42142.55	42142.56
8	Mon-Wed	20226.33	20228.44	42142.33	42144.44

Figure 3, Data from PROC IMPORT printed with formats removed.

CONCLUSION

This has been a quick look at SAS Date with decimal time. We have seen that often SAS treats the value as a regular date but other times not. We have also seen how to convert the data with decimal time to SAS Date-Time and how to make comparisons work using the INT function.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: John Henry King
Enterprise: Ouachita Clinical Data Services, Inc.
Address: 1769 AR 240
City, State ZIP: Caddo Gap, AR 71935
Work Phone: 870-356-3033
E-mail: datanull@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.